

important, and often overlooked, consideration. The Power Analysis utility allows a researcher to determine a priori the approximate sample size for a given power. Additionally, the researcher is able to quickly evaluate the power of statistical tests in the literature. The only drawback to StatChoice was the incorrect sample size estimates given for regression analyses with greater than eight variables. This problem should be easy for the authors to fix in the second version of their program. Overall, I would highly recommend this program to school psychologists who are conducting and evaluating research.

## REFERENCES

Cohen, J. (1977). *Statistical power analysis for the behavioral sciences*. New York: Academic.

# COMPUTER PROGRAMS

## *Standard Score Calculator*

Marley Watkins

South West EdPsych Services, Phoenix, Arizona

School psychologists frequently use norm-referenced test scores and often find it necessary to compare a student's performance on several tests, each of which reports results in a different standard score format. Huebner (1988, 1989) demonstrated that teachers and school psychologists may render nonequivalent special education decisions when statistically equivalent normative scores are presented in different standard score formats.

This microcomputer program was developed as a tool for school psychologists. It quickly converts a score entered in  $z$ ,  $T$ , normal curve equivalent (NCE), Wechsler IQ, Binet IQ, or Wechsler subtest format into all other formats. By automatically performing these computations, it removes some of the information processing load from the school psychologist and may contribute to improved reliability of special education decisions.

## PROGRAM DESCRIPTION

The Standard Score Calculator program was written in Applesoft BASIC for the Apple II line of microcomputers. It can easily be translated, however, to operate on most other microcomputers. After the program is typed into the computer and execution is begun, the initial input screen should appear similar to Figure 1. The user simply chooses the type of standard score to enter and then, as demonstrated in Figure 2, inputs that obtained score for conversion into other score formats. This output is presented in Figure 3. Note that the Wechsler IQ score of 107 that was originally entered has been converted into statistically equivalent  $z$ ,  $T$ , NCE, Binet, subtest, and percentile scores.

Standard Score Calculator was written in simplified style and was amply commented for easy translation into other BASIC dialects. There are consequently no error trapping, graphics, or sound routines included. Psychologists who are familiar with BASIC may wish to add error trapping and other program enhancements.

## REFERENCES

- Huebner, E. S. (1988). Bias in teachers' special education decisions as a function of test score reporting format. *Journal of Educational Research*, 21, 217-220.
- Huebner, E. S. (1989). Errors in decision-making: A comparison of school psychologists' interpretations of grade equivalents, percentiles, and deviation IQs. *School Psychology Review*, 18, 51-55.

## PROGRAM LISTING

```

2 PRINT CHR$(21): HOME
4 REM -----
5 QP$ = "Standard Score Calculator"
6 QA$ = "by Marley Watkins"
7 REM -----

9 GOTO 50000: REM
10000 REM MAIN LINE
10001 REM DISPLAY SCREEN
10010 HOME : PRINT LNS
10020 VTAB 2: HTAB 7: PRINT QP$
10030 VTAB 3: HTAB 11: PRINT QA$
10040 VTAB 4: HTAB 1: PRINT LNS
10100 FOR I = 1 TO 6
10110 VTAB I + 7: HTAB 8: PRINT I". "SS$(I)
10120 NEXT I
10200 VTAB 18: HTAB 1: PRINT "Input which type of score?";: GET A$: PRINT A$
10205 IF A$ < "1" OR A$ > "6" THEN PRINT B$: GOTO 10200
10210 A = VAL (A$)
10300 VTAB 19: HTAB 1: PRINT SS$(A)" value"
10310 VTAB 19: HTAB 26: INPUT SS: REM
11000 REM CALCULATE SCORES
11010 FOR I = 1 TO 6
11020 SS(I) = ((SS - MN(A)) / SD(A)) * SD(I) + MN(I)
11030 NEXT I
11050 REM PERCENTILE
11055 MM = MN(A):SD = SD(A)
11060 T = ABS (.7071067812 * (SS - MM) / SD)
11066 T = (1 - ((((((0.0000430638 * T + .0002765672) * T + .0001520143) * T +
.0092705272) * T + .0422820123) * T + .0705230784) * T + 1) ^ (- 16)) / 2 + .5
11067 IF SS < MM THEN T = 1 - T
11068 SS(7) = 100 * T: REM
11100 REM CHECK LIMITS
11110 FOR I = 2 TO 7
11120 IF SS(I) < 0 THEN SS(I) = 0
11130 NEXT I: REM
12000 REM DISPLAY SCORES
12010 VTAB 5: HTAB 1: CALL - 958
12020 FOR I = 1 TO 7
12030 VTAB I + 7: HTAB 5: PRINT SS$(I)":": HTAB 30: PRINT SS(I)
12035 IF I > 1 AND I < 7 AND SS(I) < 1 THEN PRINT B$: VTAB 22: HTAB 1: PRINT LNS:
VTAB 23: HTAB 4: PRINT "Extreme score. Verify your data.":: GET A$: VTAB
22: HTAB 1: CALL - 958

12040 NEXT I
12080 VTAB 23: HTAB 1: PRINT LNS:
12090 VTAB 24: HTAB 8: PRINT "Press a key to continue":: GET A$
12099 GOTO 10000: REM
50000 REM SET-UP
50010 DIM MN(6),SD(6),SS$(7)
50020 FOR I = 1 TO 40:LNS = LNS + "-": NEXT
50030 B$ = CHR$(7)
50100 MN(1) = 0:MN(2) = 50:MN(3) = 50:MN(4) = 100:MN(5) = 100:MN(6) = 10
50110 SD(1) = 1:SD(2) = 10:SD(3) = 21.06:SD(4) = 15:SD(5) = 16:SD(6) = 3
50111 SS$(1) = "z score"
50112 SS$(2) = "T score"
50113 SS$(3) = "NCE score"
50114 SS$(4) = "Wechsler IQ score"
50115 SS$(5) = "Binet IQ score"
50116 SS$(6) = "Wechsler subtest"
50117 SS$(7) = "percentile"
59998 REM GO TO MAIN LINE
59999 GOTO 10000

```

---

Standard Score Calculator  
by Marley Watkins

---

1. z score
2. T score
3. NCE score
4. Wechsler IQ score
5. Binet IQ score
6. Wechsler subtest

Input which type of score?

---

Figure 1. Initial input screen.

---

Standard Score Calculator  
by Marley Watkins

---

1. z score
2. T score
3. NCE score
4. Wechsler IQ score
5. Binet IQ score
6. Wechsler subtest

Input which type of score? 4  
Wechsler IQ score value ? 107

---

Figure 2. Second input screen.

### AVAILABILITY

The Apple II Standard Score Calculator program is available free of charge by sending a blank 5 1/4" floppy disk and an appropriate prepaid mailer to Marley Watkins, South West EdPsych Services, Inc., Phoenix, AZ 85001. The complete BASIC listing follows for those who wish to enter the program directly and for those who must modify it for operation on other microcomputers.

---

Standard Score Calculator  
by Marley Watkins

---

z score	.46666667
T score	54.6666667
NCE score	59.828
Wechsler IQ score	107
Binet IQ score	107.466667
Wechsler subtest	11.4
percentile	67.9630919

Press any key to continue

---

Figure 3. Final output screen of score conversions.